級数と微積分のルネッサンス
ーダンハムの「微積分名作ギャラリー」からー

SHIMURA Masato
JCD02773@nifty.ne.jp

2010 年 10 月 22 日

目次

概要

ダンハムの「微積分名作ギャラリー」から微積分創世記のニュートン、ライプニッツ、ベルヌイ · ファミリー、オイラーの名画を鑑賞しよう。彼等は級数の達人であって、級数を錬成しながら微積分のアイデアを得ている。ダンハムの選んだ（解説つき）名画を先端計算言語 J を用いて動かしてみよう。

ダンハムは 12 人の数学者の功績を 42 枚の絵画になぞらえて微積分の数学史を示してくれる。

創世の時代	ニュートン
	ライプニッツ
	ベルヌイ · ファミリー
	オイラー
古典	コーシー
	リーマン
	リュービュ
	ワイエルシュトラウス
現代	カントール
	ヴォルテラ
	ベール
	ルベーグ

A) 音楽を学ぶ人はバロック時代のバッハや古典のモーツアルト、ベートーベンをレパートリーとするし、絵画を学ぶ人はルネッサンスの巨匠やバロックのレンブラントなど古典を最初に模写する。数学でも古典をしっかりと学んで欲しいと言うのが W· ダンハムの考えである。

Z) 数学の専門家でないとあまり馴染みのない名前もあるし、何故入っていないのと思う人も多いが、皆なかなか手強い

A）微分や積分はアイルランドのバークレー (1685-1753) 司教に異端とまで言われ強烈な批判に晒され、この暗雲は 250 年近く尾を引いた。数学者の苦闘の歴史でもあり、体調を崩した人も少なからずいた。

1　ニュートン (1642-1727)

A) この時代の数学者は級数の達人だ。級数が先にあって微分は後から出来た。

Z）テイラー (1685-1731) やマクローリン (1698-1746) はニュートンの弟子では？

A)2 人はニュートンとライプニッツの大英帝国と大陸派との熾烈な微積分先取論争のニュートン派の騎士だ。級数展開は当時広く知られていた。

1.1　一般 2 項展開

一般 2 項展開 (-1665)

$$(P+PQ)^{\frac{m}{n}} = P^{\frac{m}{n}} + \frac{m}{n}AQ + \frac{m-n}{2n}BQ + \frac{m-2n}{3n}CQ + \frac{m-3n}{4n}DQ + \cdots \tag{1}$$

2 項式を級数に展開する。$(P+PQ)^{\frac{m}{n}}$ を $\dfrac{m}{n}$ が整数でも分数でも、正でも負でも級数に展開する

Newton による Example

$$\sqrt{c^2+x^2} = \left(c^2 + c^2\left(\frac{x^2}{c^2}\right)\right)^{\frac{1}{2}}$$

$$\begin{cases} P = c^2 & Q = \dfrac{x^2}{c^2} \\ m = 1 & n = 2 \end{cases}$$

とする。(1) に代入する。$m < n$ なのでマイナスが生じる。

$$\sqrt{c^2+x^2} = (c^2)^{\frac{1}{2}} + \frac{1}{2}A\frac{x^2}{c^2} - \frac{1}{4}B\frac{x^2}{c^2} - \frac{3}{6}C\frac{x^2}{c^2} - \frac{5}{8}D\frac{x^2}{c^2} - \cdots$$

A,B,C,D はそれぞれ一つ前の項の値であるので逐次更新する
$A = (c^2)^{\frac{1}{2}} = c$ である。$\dfrac{c}{2}\cdot\dfrac{x^2}{c^2} = \dfrac{x^2}{2c}$ から

$$\sqrt{c^2+x^2} = c + \frac{x^2}{2c} - \frac{1}{4}B\frac{x^2}{c^2} - \frac{1}{2}C\frac{x^2}{c^2} - \frac{5}{8}D\frac{x^2}{c^2} - \cdots$$

$B = \dfrac{x^2}{2c}$ である

$$\sqrt{c^2+x^2} = c + \frac{x^2}{2c} - \frac{x^4}{8c^3} - \frac{1}{2}C\frac{x^2}{c^2} - \frac{5}{8}D\frac{x^2}{c^2} - \cdots$$

$C = -\dfrac{x^4}{8c^3}$ である
$D = \dfrac{x^6}{16c^5}$ である

$$\sqrt{c^2+x^2}=c+\frac{x^2}{2c}-\frac{x^4}{8c^3}+\frac{x^6}{16c^5}-\frac{5x^8}{128c^7}-\cdots$$

---

(1) の両辺を $P^{\frac{m}{n}}$ で割る(これで今日の形になる）

$$(1+PQ)^{\frac{m}{n}}=1+\frac{m}{n}Q+\frac{\frac{m}{n}\left(\frac{m}{n}-1\right)}{2\times 1}Q^2+\frac{\frac{m}{n}\left(\frac{m}{n}-1\right)\left(\frac{m}{n}-2\right)}{3\times 2\times 1}Q^3+\cdots \tag{2}$$

Newton による Example

'bar' plot newton_ex steps 0 0.99 20



$$\frac{1}{\sqrt{1-x^2}}=(1-x^2)^{-\frac{1}{2}}$$

右項から係数を抜き出す

$$\begin{cases} Q=-x^2 \\ m=-1 \quad n=2 \end{cases} \text{とする}$$

(2) に代入する

$$\frac{1}{\sqrt{1-x^2}}=1+\left(-\frac{1}{2}\right)(-x^2)+\frac{\left(-\frac{1}{2}\right)\left(-\frac{3}{2}\right)}{2\times 1}(-x^2)^2+\frac{\left(-\frac{1}{2}\right)\left(-\frac{3}{2}\right)\left(-\frac{5}{2}\right)}{3\times 2\times 1}(-x^2)^3+\frac{\left(-\frac{1}{2}\right)\left(-\frac{3}{2}\right)\left(-\frac{5}{2}\right)\left(-\frac{7}{2}\right)}{4\times 3\times 2\times 1}(-x^2)^4\cdots$$

$$=1+\frac{1}{2}x^2+\frac{3}{8}x^4+\frac{5}{16}x^6+\frac{35}{128}x^8+\cdots \tag{3}$$

■J で計算　分数部分の係数を求めるスクリプトを作成する

```
2 5 $ 10  bin_newton_sub _1r2
     1       1r2       3r8        5r16       35r128
63r256 231r1024 429r2048 6435r32768 12155r65536
```

```
'bar' plot | 20 bin_newton_sub _1r2
```



```
2 5 $ 10 bin_newton_sub _1r2
      1      _1r2         3r8       _5r16         35r128
_63r256 231r1024 _429r2048 6435r32768 _12155r65536
```

A) 関数の振る舞いを楽しんでいるようだ池の鯉を見るように

■Script

```
bin_newton_sub =: 4 : 0
NB. generalized binomial series by Newton
NB. x =jisuu ex. 4
NB. y beki _1r2
NB. Usage: 10 bin_newton_sub _1r2
bunsi=: ; */ L:0 ({@> x # y)- L:0 <\ i. x
| 1,1 x:  bunsi * %  ! >: i.x
)
```

■検算 .

$$\left(1+\frac{1}{2}x^2+\frac{3}{8}x^4+\frac{5}{16}x^6+\frac{35}{128}x^8\right)\times\left(1+\frac{1}{2}x^2+\frac{3}{8}x^4+\frac{5}{16}x^6+\frac{35}{128}x^8\right)$$

「上は次のようになるからやって見よと」（ダンハム）

$$=1+x^2+x^4+x^6+x^8+\cdots$$

やってみると x^{10} の次の項から残余がでるが次数を増やせば次第に消える

```
a1,+/a1=. (-i.5)|."0 1 (a*/a=. 4 bin_newton_sub _1r2),. 5 5 $ 0
1 1r2 3r8 5r16 35r128      0        0        0          0 0 NB. 初項
0 1r2 1r4 3r16   5r32 35r256        0        0          0 0
0   0 3r8 3r16   9r64 15r128 105r1024        0          0 0
0   0   0 5r16   5r32 15r128   25r256 175r2048          0 0
0   0   0    0 35r128 35r256 105r1024 175r2048 1225r16384 0
1   1   1    1      1 65r128  155r512 175r1024 1225r16384 0 NB. sum
```

$1+x^2+x^4+x^6+x^8+\cdots$ は公比 x^2 の無限等比級数で和は $\dfrac{1}{1-x^2}$ となるので $1+x^2+x^4+x^6+x^8+\cdots$ の 2 乗が $\dfrac{1}{1-x^2}$ なら級数自体は $\dfrac{1}{\sqrt{1-x^2}}$ である

A) ダンハムがニュートンに「お見事」と拍手している

■J Grammar　.

#	*copy*	指定個数複写する
!	*factorial*	階乗
>:	*add*	+1
$L:0$	*Levelat*	ボックス内の演算
<\	*prefix*	一個づつずらした組合せを作る
|	*absolute*	絶対値
*/	*times*	Π関数
|.	*rotate*	回転
"01	*rank*	01 はボックスに縦横に作用する

steps 0 1 100

Δx の数列を生成する.

from 0 to 1 divide by 100

`numeric.ijs` に入っている

`require 'plot numeric'`

1.2　級数の逆関数

級数の逆関数

$$x = z - z^2 + z^3 - z^4 + z^5 + \cdots$$

は次の関数の逆関数である

$$z = x - x^2 + x^3 - x^4 + x^5 + \cdots$$

ここから始める

$$z = x - x^2 + x^3 - x^4 + x^5 + \cdots \tag{4}$$

次のように書き直す

$$(x - x^2 + x^3 - x^4 + x^5 + \cdots) - z = 0$$

2 次以降の高次の項を捨てる

$$x - z = 0 \to x = z$$

正確な解を $x = z + p$ として (4) に代入する

$$[(z+p) - (z+p)^2 + (z+p)^3 - (z+p)^4 + (z+p)^5 + \cdots] - z = 0$$

A) ニュートンはこのように多段に接木をし、継いだ穂先の根本の一節だけ用いて後は捨ててしまう手法が得意だ。

Z)n 乗の計算はパスカルの 3 角形を用いよう

```
 pascal 4
1 0 0 0 0
1 1 0 0 0
1 2 1 0 0
1 3 3 1 0
1 4 6 4 1
```

```
pascal=:3 : '|:!/~i.>:y'
```

$$\begin{array}{c|ccccc|ccccc}
- & 1 & & & & & z & & & & \\
+ & 1 & 1 & 0 & 0 & 0 & z & p & & & \\
- & 1 & 2 & 1 & 0 & 0 & z^2 & 2zp & p^2 & & \\
+ & 1 & 3 & 3 & 1 & 0 & z^3 & 3z^2p & 3zp^2 & p^3 & \\
- & 1 & 4 & 6 & 4 & 1 & z^4 & 4z^3p & 6z^2p^2 & 4zp^3 & p^4
\end{array}$$

縦に整理すると

$$\begin{gathered}(-z^2 + z^3 - z^4 - \cdots) + (1 - 2z + 3z^2 - 4z^3 + \cdots)p \\ +(-1 + 3z - 6z^2 + \cdots)p^2 + (1 - 4z + \cdots)p^3 + (-1 + \cdots)p^4\end{gathered} \tag{5}$$

p の 2 次以降の項を切り捨て、p で整理する[*1]

$$p = \frac{z^2 - z^3 + z^4 - \cdots}{1 - 2z + 3z^2 - 4z^3 + \cdots}$$

最低次の項のみ残して大胆に捨てる

[*1] 分子の符号は逆になる

$$p = \frac{z^2}{1}$$

ここで逆関数は $x = z + p = z + z^2$ まで算出できた

$p = z^2 + q$ として q を求める。(5) に代入する

$$\begin{gathered}(-z^2 + z^3 - z^4 - \cdots) + (1 - 2z + 3z^2 - 4z^3 + \cdots)(z^2 + q) \\ +(-1 + 3z - 6z^2 + \cdots)(z^2 + q)^2 + (1 - 4z + \cdots)(z^2 + q)^3 + (-1 + \cdots)(z^2 + q)^4\end{gathered} \tag{6}$$

q で整理する

$$\begin{gathered}(-z^3 + z^4 - z^6 + \cdots) + (1 - 2z3z^2 + 2z^3 - \cdots)q \\ +(-1 + 3z - 3z^2 - 2z^3 + \cdots)q^2 + \cdots\end{gathered} \tag{7}$$

q で整理する

$$q = \frac{z^3 - z^4 + z^6 - \cdots}{1 - 2z + z^2 + 2z^3 + \cdots}$$

分子分母とも最低次以外を捨てる

$$q = \frac{z^3}{1}$$

ここで逆関数は $x = z + p = z + z^2 + z^3$ まで算出できた

以降繰り返す

$$x = z + z^2 + z^3 + z^4 + z^5 + \cdots$$

1.2.1　検算

$z = x - x^2 + x^3 - x^4 + x^5 + \cdots$ は等比級数で公比は $-x$

$$z = \frac{x}{1 + x} \to x = \frac{z}{1 - z}$$

$$= z - z^2 + z^3 - z^4 + z^5 + \cdots$$

1.3　求積



$$\int_0^x at^{\frac{m}{n}} dt = \frac{ax^{(\frac{m}{n})+1}}{(\frac{m}{n})+1} = \frac{an}{m+n} x^{(\frac{m+n}{n})} \tag{8}$$

Z）ニュートンはナイフ 1 本にハンカチを被せてマジックを披露するが

A) ニュートンの中心テーマは**流率**でこれは軍団仕様。微積分はデザートかな

1.3.1　ニュートンの証明

ニュートンは (8) に証明を与えている。



$$z(x) = \frac{an}{m+n} x^{\frac{m+n}{n}}$$

$$\begin{cases} c = & \dfrac{an}{(m+n)} \\ p = & m+n \end{cases}$$ とおく

$$z(x) = cx^{\frac{p}{n}}$$

$$(z(x))^n = c^n x^p \tag{9}$$

A) ナイフ 1 本を添える。

$z(x+o) = z(x) + ov$ として (9) に代入する

$$(z(x)+ov)^n = (z(x+v))^n = c^n(x+o)^p$$

両辺を 2 項展開する

$$(z(x))^n + n(z(x))^{n-1}ov + \frac{n(n+1)}{2}(z(x))^{n-2}o^2v^2 + \cdots$$

$$= c^n x^p + c^n p x^{p-1} o + c^n \frac{p(p-1)}{2} x^{p-2} o^2 + \cdots$$

両辺の第 1 項を消す。→　o で割る

$$n(z(x))^{n-1}v + \frac{n(n+1)}{2}(z(x))^{n-2}ov^2 + \cdots$$

$$= c^n p x^{p-1} + c^n \frac{p(p-1)}{2} x^{p-2} o + \cdots$$

o のある項を消す

$$n(z(x))^{n-1}y = c^n p x^{p-1} \tag{10}$$

(10) から y を求める

$$y = \frac{c^n p x^{p-1}}{n(z(x))^{n-1}} = \frac{\left(\frac{an}{(m+n)}\right)^n (m+n)x^{m+n-1}}{n\left(\frac{an}{(m+n)}x^{\frac{(m+n)}{n}}\right)^{n-1}} = ax^{\frac{m}{n}}$$

Z) ニュートンの頭脳に青白い稲妻が走ったようだが、マジックを見ているようだ。
A) 流率の極意らしい

1.4　*arcsin* の級数展開 (1669)

$$arcsin = z = x + \frac{1}{6}x^3 + \frac{3}{40}x^5 + \frac{5}{112}x^7 + \frac{35}{1152}x^9 + \cdots$$

---

(2) を用いて $\dfrac{1}{\sqrt{1-x^2}}$ を展開する

$$dz = [1 + \frac{1}{2}x^2 + \frac{3}{8}x^4 + \frac{5}{16}x^6 + \frac{35}{128}x^8 + \cdots]dx \tag{11}$$

$$z = x(z) = \int_0^x dz = \int_0^x [1 + \frac{1}{2}t^2 + \frac{3}{8}t^4 + \frac{5}{16}t^6 + \frac{35}{128}t^9 + \cdots]dt$$

積分して

$$arcsin = z = x + \frac{1}{6}x^3 + \frac{3}{40}x^5 + \frac{5}{112}x^7 + \frac{35}{1152}x^9 + \cdots \tag{12}$$

```
require 'plot numeric trig'
 'key sin arcsin'  plot _0.99 0.99;'sin,arcsin'
```



---

arcsin から *sin* を求める

$$sin(z) = z - \frac{1}{6}z^3 + \frac{1}{120}z^5 - \frac{1}{5040}z^7 + \frac{1}{362880}z^9 + \cdots$$

$$= \sum_{k=0}^{\infty} \frac{(-1)^k}{(2k+1)!} z^{2k+1}$$

A)*sin* の逆関数 *arcsin* の方が歴史上先に求められた

Z) ニュートンがヨーロッパの文献では最初に *sin,cos* の級数展開を記述した。オリジンはインドで更に 200 年遡る

```
   ! 1 3 5 7 9
1 6 120 5040 362880
```

$$cos(z) = \sum_{k=0}^{\infty} \frac{(-1)^k}{(2k)!} z^{2k}$$

これらの級数の収束は非常に早い。

```
 % (5$1 _1)* ! 1 3 5 7 9   NB. sin
1 _0.166667 0.00833333 _0.000198413 2.75573e_6
```

```
'bar' plot % (5$1 _1)* ! 1 3 5 7 9
```



初項を x とし、$x = z + p$ とおいて (12) に代入する

$$sin(z) = z - \frac{1}{6}(z+p)^3 + \frac{1}{40}(z+p)^5 - \frac{1}{112}(z+p)^7 + \frac{1}{1152}(z+p)^9 + \cdots$$

```
 1 3 5 7 9{  pascal 9
1 1  0  0   0   0  0  0 0 0
1 3  3  1   0   0  0  0 0 0
1 5 10 10   5   1  0  0 0 0
1 7 21 35  35  21  7  1 0 0
1 9 36 84 126 126 84 36 9 1
```

$$\begin{bmatrix} z & p & & & & & & & & \\ z^3 & 3z^2p & 3zp^2 & z^3 & & & & & & \\ z^5 & 5z^4p & 10z^3p^2 & 10z^2p3 & 5zp^4 & z^5 & & & & \\ z^7 & 7z^6p & 21z^5p^2 & 35z^4p^3 & 35z^3p^4 & 21z^2p^5 & 7zp^6 & z^7 & & \\ z^9 & 9z^8p & 36z^7p^2 & 84z^6p^3 & 126z^5p^4 & 126z^4p^5 & 84z^3p^6 & 36z^2p^7 & 9zp^8 & z^9 \end{bmatrix}$$

$$p = \frac{-\frac{1}{6}z^3 - \frac{1}{40}z^5 - \frac{1}{112}z^7 - \cdots}{1 + \frac{3}{6}z^2 + \frac{5}{40}z^4 + \frac{7}{112}z^6 \cdots}$$

分子、分母とも最少次数の項をとる

$$p = -\frac{1}{6}z^3$$

$$x = z - \frac{1}{6}z^3 + \cdots$$

to be continued

1.5　ニュートン派と数値計算

A) ダンハムのニュートンの名画は以上であり、彼の選んだ名画は以後は大陸系のみである。ニュートンの後継者達は数値計算や級数展開で大きな功績を残しているし、微積分の先取論争でニュートンの騎士団として奮戦している。

1.5.1　ニュートン · ラフソン法

ニュートン法またはニュートン · ラフソン法と呼ばれる非線形方程式の解法がある。

ニュートンは唯一つ次の式の解法のメモを残した。

$y^3 - 2y - 5 = 0$

ニュートンは第 4 近似で 2.09455147 の解を得ている

■Newton の行った手計算　.

解の点の値を $y = 2 + p$ とおく

$(2 + p)^3 - 2(2 + p) - 5 = 0$

$-1 + 10p + 6p^2 + p^3 = 0$

p^2, p^3は微少故に無視すると

$-1 + 10p = 0 \rightarrow P = 0.1$

$y = 2 + 0.1 = 2.1$　（第一近似）

$P = 0.1 + q$ とおく

$-1 + 10(0.1 + q) + 6(0.1 + q)^2 + (0.1 + q)^3$

$= 0.061 + 11.23q + 6.3q^2 + q^3$

q^2, q^3は微少故に無視すると

$11.23q = -0.061 \rightarrow q = -0.00543188$

$y = 2 + 0.1 - 0.00543188 = 2.09457$(第 2 近似)

$q = -0.00543188 + r$ とおく

$0.061 + 11.23(-0.00543188 + r) + 6.3(-0.00543188 + r)^2 + (-0.00543188 + r)^3$

$= 0.000185723 + 11.1616r + 6.2837r^2 + r^3$

r^2, r^3は微少故に無視すると

$11.1616r = 0.000185723 \rightarrow r = -0.0000166395$

$y = 2 + 0.1 - 0.00543188 - 0.0000166395 = 2.09455$(第 3 近似)

コンピュータは通常はこの辺りで収束と判断するが更に続けると

$r = -0.0000166395 + s$ とおく

$0.000185723 + 11.1616(-0.0000166395 + s) + 6.2837(-0.0000166395 + s)^2 + (-0.0000166395 + s)^3$

$= 0.000371448 + 11.1618s + 6.28375s^2 + s^3$

s^2, s^3は微少故に無視すると

$11.1618s = 0.000371448 \rightarrow s = -0.0000332785$

$y = 2 + 0.1 - 0.00543188 - 0.0000166395 - 0.0000332785 = 2.09455$(第 4 近似)

s^2, s^3 の項を大胆に捨てているが収束は非常に早い。

■多項式 p. .

多項式の解は J のプリミティブ p. を用いて直ちに得られる。

$y = -5 - 2x + x^3$

```
 p. _5 _2 0 1
+-+------------------------------------------+
|1|2.09455 _1.04728j1.13594 _1.04728j_1.13594|
+-+------------------------------------------+
```

[*2]

もう少し先まで表示してみる。文字化 (":) を用いる。

```
0j15 ":  {. ;{: p. _5 _2 0 1
2.094551481542327
```

[*3]

[*2] 最初のボックスは反復数

[*3] J は数値のどれかに x を付けると浮動小数点でなく分数を用いて細密に計算する。内容は x:に 1,2,_1,_2 のパラメーターを付けて見ることができる。

```
 % 3x
1r3  NB. simple

 _1 x: % 3x
0.333333
```

■ニュートン · ラフソン法 *Joseph Raphson*(1648 – 1715)

1690 年に出版された著書に *Newton – Raphson method* が載っている。ケンブリッジ出身でハーレーの推薦で Royal Society のメンバーに。対ライプニッツ論争の騎士の一人

ニュートン法は、

$$x - \frac{f(x)}{f'(x)}$$

の反復計算を行うものである。J は微分演算子 D. があるので APL より容易にスクリプトが作れる。次のスクリプトは J の定番である。

ニュートン法は微分演算子を用いてシンプルに定義できる。動詞を左パラメーターに取るので副詞で定義しなければならない。ランクはベクトルを引数に取ることができるように `("0)` とする。`(^:_)` は収束まで計算するようにしているが `(^:100)` 程度で打ち切っても良い。

Example $y = -5 - 2x + x^3$ をニュートン法で解いてみよう。

多項式は p. を用いて定義する。

```
 2 5 $  _5 _2 0 1&p. new_1 i:5
2.09455 2.09455 2.09455 2.09455 2.09455
2.09455 2.09455 2.09455 2.09455 2.09455
```

```
NB. Newton method
 new_1=: 1 : ' ] - x % x D.1' (^:_)("0)
```

```
f=: _5 _2 0 1&p.
plot _5 5;'f ,f new_1'
```



1.5.2 テーラー展開の数値計算

Brook Taylor(1685 – 1731) 父母共にイングランドの名門。ケンブリッジの St. John College で数学と古典を学ぶ。1712 Royal Society のメンバーに選ばれ, 間もなく対ライプニッツ論争の英国委員会のメンバーや Royal Society の書記長になる。

A) 級数展開は当時ポピュラーであったが、テーラーも研究している。テーラー展開の名誉は

```
0j10 ": _1 x: % 3x
0.3333333333
```

ニュートン派の選手団長テーラーに与えよう。

■J のテーラー展開　J はテーラー展開のプリミティブとして t. t: T. を持っている

- 階乗

```
 ! i.8   NB. 階乗
1 1 2 6 24 120 720 5040
1 x: % ! i.8  NB. 分数表示
1 1 1r2 1r6 1r24 1r120 1r720 1r5040
```

- t.

```
 1 x: (^ t. i.8), (1&o. t. i.8),: 2&o. t. i.8
1 1  1r2  1r6 1r24 1r120  1r720  1r5040 NB. e
0 1    0 _1r6    0 1r120      0 _1r5040 NB. sin
1 0 _1r2    0 1r24     0 _1r720       0 NB. cos
```

- t: テーラー展開の係数 (± を求める)

```
 (^ t: i.8), (1&o. t: i.8),: 2&o. t: i.8
1 1  1  1 1 1  1  1 NB. e
0 1  0 _1 0 1  0 _1 NB. sin
1 0 _1  0 1 0 _1  0 NB. cos
```

- T: テーラー展開の次数を指定する

```
 ^ T. 8 i.8   NB. 8 次
1 2.71825 7.38095 19.8464 51.8063 128.619 300.143 656.569


 ^ T.  5 i.8   NB。5 次
1 2.70833 7 16.375 34.3333 65.375 115 189.708
```

■和算の級数展開　上方や江戸の和算家は π の計算に級数を用いた。

建部賢弘（1710) $\left(\frac{sin^{-1}x}{2}\right)^2$ の級数に $x = \frac{1}{2}$ を代入したもの

$$\pi^2 = 9\left(1 + \frac{1^2}{3 \cdot 4} + \frac{1^2 + 2^2}{3 \cdot 4 \cdot 5 \cdot 6} + \frac{1^2 + 2^2 + 3^2}{3 \cdot 4 \cdot 5 \cdot 6 \cdot 7 \cdot 8} + \cdots\right)$$

松永良弼（1739) $sin^{-1}x$ の級数に $x = \frac{1}{2}$ を代入したもの

$$\pi = 3\left(1 + \frac{1^2}{4 \cdot 6} + \frac{1^2 + 3^2}{4 \cdot 6 \cdot 8 \cdot 10} + \frac{1^2 + 3^2 + 5^2}{4 \cdot 6 \cdot 8 \cdot 10 \cdot 12 \cdot 14} + \cdots\right)$$

■**J で計算**　建部級数　収束が早い (左項は π とした)

```
   tatebe_series0 12
+-------+-----------------------------------------------------+
|3.14347|1 0.0833333 0.0138889 0.000694444 1.65344e_5 2.29644e_7|
+-------+-----------------------------------------------------+
```

松永級数も収束は早い

```
   matunaga_series0 8
+-------+-----------------------------------------------------------------------------+
|3.14095|1 0.0416667 0.00520833 0.000108507 9.04225e_7 4.03672e_9 1.12131e_11 2.12369e_14|
+-------+-----------------------------------------------------------------------------+
```

■Script

```
tatebe_series0=: 3 : 0
NB. u 12 //even number
A=:  ({@> }.i. y ){. L:0  }.i. y NB. bunsi
B=: ({@>}. +: i. -: y){. (L:0) 3}.  i. +: y NB. bunbo
A=: (#B){. A
TMP=: 1,(;+/@:*: L:0 A) %  ; */ L:0 B
 ( %: 9*+/ TMP);TMP
)

matunaga_series0=: 3 : 0
NB. u 12 //even number
A=:  ({@> }.i. y ){. L:0 >: +: i. y NB. bunsi
B=: ((# tmp)$ 0 1 ) # tmp=. <\ 2}.  +: i. +: y NB. bunbo
A=: (#B){. A
TMP=: 1,(;+/@:*: L:0 A) %  ; */ L:0 B
 ( 3*+/ TMP);TMP
 )
```

■J Grammar　.

{@>		数値を 1 個づつボックスで囲み $L:0$ で並列演算
$+:$	$Double$	2 倍
$-:$	$Half$	半分
$i.$	$Integer$	0 から指定数までの整数列を生成
$+/$	Sum	合計

1.5.3　シンプソン積分

$Thomas\ Simpson(1710-1761)$

父は織工で、トーマスは織工からスタートし、正規の教育は初等教育のみで数学は自習した。15 才で下級学校の教師。ダービーで星占術で問題を起こし、ロンドンへ。最初コーヒーハウス（大衆向きの教室でもあった）で教えていたが、1743 年に王立の Militaly Academy の数学のヘッドに就任し、2 年後に Royal Society のフェローに選ばれた。シンプソン法は自身の知識とニュートンに幾らか学んで生まれたもの。ニュートン · ラフソン法に微分を用いて今日の形にしたのもシンプソンである。



(x_2 が現在のポジションとする)

A) 積分の数値計算の世界ではシンプソン法がクラシックの名作であり簡潔なアルゴリズムと相応な精度を有している。

Z) 大御所リーマンは台形則としてぞんざいに扱われている

■シンプソンのアルゴリズム　3 点 $(x_1, y_1), (x_2 = \dfrac{x_1 + x_3}{2}), (x_3, y_3)$ を通る 2 次関数の多項式（中心差分）

$$f(x) = ax^2 + bx + c$$

$$\begin{cases} ax_1^2 + bx_1 + c & = y_1 \\ ax_2^2 + bx_2 + c & = y_2 \\ ax_3^2 + bx_3 + c & = y_3 \end{cases}$$

マトリクスで表すと

$$\begin{bmatrix} x_1^2 & x_1 & 1 \\ x_2^2 & x_2 & 1 \\ x_3^2 & x_3 & 1 \end{bmatrix} \begin{bmatrix} a \\ b \\ c \end{bmatrix} = \begin{bmatrix} y_1 \\ y_2 \\ y_3 \end{bmatrix}$$

数式処理で解くと（平野による）

$$a = \frac{2(y_1 - 2y_2 + y_3)}{(x_1 - x_3)^2}$$

$$b = \frac{x_3(3y_1 - 4y_2 + y_3) + x_1(y_1 - 4y_2 + 3y_3)}{(x_1 - x_3)^2}$$

$$c = \frac{x_3(x1 + x_3)y_1 - (4x_1x_3y_2) + x_1(x_1 + x_3y_3)}{(x_1 - x_3)^2}$$

a, b, c を $f(x) = ax^2 + bx + c$ に代入して区間 $[x_1, x_3]$ で積分するとシンプソンの公式が現れる

$$\int_1^3 f(x)dx = \frac{1}{6}(x_3 - x_1)(y_1 + 4y_2 + y_3)$$

$(x_1 - x_3)$ は中心差分で $2\Delta x$ であるから (x_1, x_3) を a, b とすると

$$I(f) = \frac{\Delta x}{3}\left(f(a) + 4f\left(\frac{a+b}{2}\right) + f(b)\right) = \frac{b-a}{6}\left(f(a) + 4f\left(\frac{a+b}{2}\right) + f(b)\right)$$



Z) これは移動平均ですね

A) 次の式の方が良く用いられる

$$I(f) = \frac{1}{3}\left(f(x_0) + 4\sum_{j=1}^{m} f(x_{2j-1}) + 2\sum_{j=1}^{m-1} f(x_{2j}) + f(x_{2m}) \right)$$

■J で計算　関数（動詞）を引数とすると一々スクリプトを書き直さなくとも良い

Example

$$\int_3^4 x^3$$

```
   ^&3 simpson0 3 4 10000  NB. 積分区間と分割数
43.7455

   ^&3 simpson0 3 4 1000000
43.75
```

スクリプトは移動平均のアルゴリズムで作成した。Exact は $\int_3^4 x^3 = \dfrac{x^4}{4} = 43.75$ である。計算精度は Δ を相当小さくしなければならず、計算限界に近寄る。

```
simpson0=: 1 : 0
'A0 B0 N0'=. y
xn =: A0, A0+ (B0-A0)* (>: i. N0) % N0
HX=: (% N0 )  % 6         NB. delta
 +/  HX * ; +/ (L:0) 1 4 1 *(L:0) u(L:0) 3<\ xn
)
```

J の `system/packages/math/integrat.ijs` にはシンプ村積分以外にも各種の積分のスクリプトが入っている

■J Grammar　.

^&3	x^3	&は数値と関数を接続する
%	*divide*	割る
u		左の動詞（関数）を引数とする

2　ライプニッツ

$Leipnitz(1646-1716)$

2.1　変形定理

$$\int ydx = \frac{1}{2}\int zdx + \frac{1}{2}by(b) - \frac{1}{2}ay(a) \qquad (13)$$

ライプニッツが積分の考えに至った変形定理。当時は曲線の下部面積を求める課題が流行っていた。次のステップを進む

Z）随分派手にフォークを並べたものだ

1. 数個の三角形の相似を考察して扇型の面積を求める
   - 直角三角形 (I),(II),(III) が相似



- $\triangle(I),(II),(III)$ は直角三角形
- 右の $\triangle(I)$ は $\dfrac{dy}{dx}$
- P において接線 t を引く
- $\triangle(I)$ と　$\triangle(II)$ ($\triangle PTD$) は相似
- $\angle TDP$ は直角である。
- $\triangle DTP$ を裏返し、$90^o + \angle PTD = \alpha$ 分回転させる
- $\angle WTO$ は $\pi - \angle PTD$
- ($\angle TWO$ は直角)
- $\angle PTD$=$\angle TOW = \alpha$

- $\triangle(II)$ と $\triangle(III)$ は相似





最初の微少な $\triangle$ は $\triangle PTD$ に写されている
$\dfrac{dy}{dx} = \dfrac{PD}{TD} = \dfrac{y-z}{x}$ から

$$z = y - x\frac{dy}{dx} \tag{14}$$

A) ライプニッツ記号が有利なところ。（ここが肝要）更に、dx は $\triangle TWO$ の $WO = h$ に、ds は $TO = z$ に写されている。

$$\frac{ds}{dx} = \frac{z}{h} \Longrightarrow hds = zdx$$



O から t に平行に補助線を引く
P から Y 軸に平行に補助線まで引いた線は z に等しいので、$dx \times z$ が得られる。
$\triangle POQ$(一個のフォーク) の面積は $\dfrac{1}{2}hds$ であり、$\dfrac{1}{2}zdx$ に等しい

$$\text{扇型 } OAB \text{ の面積} = \text{フォークの面積の和} = \int \frac{1}{2}zdx = \frac{1}{2}\int zdx$$

2. 曲線 AB と a, b に囲まれた面積を求める
   - 扇型 OAB の下部の直線は $\triangle ObB$ の面積の $\frac{1}{2}$ の線
   - $\triangle OaA$ は扇型の不要な部分と $\triangle ObB$ の不要な部分の合計になる。これ引けば扇型の下部の面積が求まる



Z) ギリシャ人はこのおいしい皿を食べ残したのか

A) スパイス $\dfrac{dy}{dx}$ が無かった。ゆっくり味わおう

Z) でもこれで計算しろと言われたらきつい

A) 数値計算でライプニッツ法云々は耳にしない。

2.2　ライプニッツ級数

ライプニッツ級数

$$\frac{\pi}{4} = 1 - \frac{1}{3} + \frac{1}{5} - \frac{1}{7} + \frac{1}{9} - \cdots \tag{15}$$

左は $\dfrac{\pi}{4}$ で右の図は y を z に変換したもの。右の図で曲線より下の部分の面積は $\dfrac{\pi}{4}$ に成ればよい



円の方程式は

$$x^2 + y^2 = 2x$$

であるので図は $\dfrac{\pi}{4}$ で四分円
微分すると

$$2xdx + 2ydy = 2dx$$

変形すると

$$\frac{dy}{dx} = \frac{1-x}{y}$$

から次の式が得られる

$$z = y - x\frac{dy}{dx} = y - x\left(\frac{1-x}{y}\right) = \frac{y^2 + x^2 - x}{y} = \frac{2x - x}{y} = \frac{x}{y}$$

*4
上式を 2 乗する

$$z^2 = \frac{x^2}{y^2} = \frac{x^2}{2x - x^2} = \frac{x}{2-x}$$

x について解く

$$x = \frac{2z^2}{1+z^2} \tag{16}$$

四角形の面積は右の図の曲線の上下に分離でき、求める面積は下の部分

$$\int_0^1 zdx = 1 - \int_0^1 xdx \tag{17}$$

(17),(16) から

$$\frac{\pi}{4} = \frac{1}{2}xy|_0^1 + \frac{1}{2}\int_0^1 \left(1 - \int_0^1 xdx\right)$$

$$= 1 - \frac{1}{2}\int_0^1 \frac{2z^2}{1+z^2}dz = 1 - \int \frac{z^2}{1+z^2}dz$$

整理して

$$\frac{z^2}{1+z^2} = z^2\left[\frac{1}{1+z^2}\right] = z^2[1 - z^2 + z^4 - z^6 + \cdots] = z^2 - z^4 + z^6 - z^8 + \cdots$$

積分して

*4 $y^2 = 2x - x^2$

$$\frac{\pi}{4} = 1 - \int_0^1 [z^2 - z^4 + z^6 - z^8 + \cdots] dz = 1 - \left[\frac{z^3}{3} - \frac{z^5}{5} + \frac{z^7}{7} - \frac{z^9}{9} + \cdots \Big|_0^1 \right]$$

そして

$$\frac{\pi}{4} = 1 - \frac{1}{3} + \frac{1}{5} - \frac{1}{7} + \frac{1}{9} - \cdots$$

```
   ls_sub 10
1 _3 5 _7 9

   leipnitz_series 10000
+----------------------+----------------------+
|3.14139265359179330000|0.78534816339794833000|
+----------------------+----------------------+
 pi                          pi/4
```

ライプニッツ級数の収束は非常に遅い

```
 plot  {. |:  ;("1) > leipnitz_series L:0 {@> >:i.300
```



$-\frac{1}{3} + \frac{1}{5} - \frac{1}{7} + \frac{1}{9} - \cdots$ は美しい曲線となる。1 は全積分区間 (0 1) の正方形であり、級数は ± しながら z に近づく。

```
a,.". 8j4 ": |: (|+/ tmp) ,~ tmp=.Lx a=.steps 0 1 10

t -(t^3)/3 (t^5)/5 -(t^7)/7 (t^9)/9 sum
  0        0       0        0       0      0
0.1 _0.0003       0        0       0 0.0003
0.2 _0.0027 0.0001        0       0 0.0026
0.3  _0.009 0.0005        0       0 0.0085
0.4 _0.0213  0.002 _0.0002       0 0.0195
```

```
0.5 _0.0417 0.0063 _0.0011 0.0002 0.0363
0.6  _0.072 0.0156  _0.004 0.0011 0.0593
0.7 _0.1143 0.0336 _0.0118 0.0045  0.088
0.8 _0.1707 0.0655   _0.03 0.0149 0.1202
0.9  _0.243 0.1181 _0.0683  0.043 0.1502
  1 _0.3333    0.2 _0.1429 0.1111 0.1651
```



```
tmp=. Lx a=. steps  1 0 10
plot a;(|+/ tmp) ,~| tmp
```



■Scripts .

```
leipnitz_series=: 3 : 0
NB. Leipnits series  is 1p1/4=1+1r3-1r5+1r7-1r9
NB. Usage: u 1000
NB. find pi /convergence is very slow
(4 * LS) ; LS=.+/ % ls_sub y
)
ls_sub=:  3 : '((#tmp)$1 _1)*; tmp=.{.(L:0)_2<\>:i.y '

 L1=: 3 : '-(y^3)%3'
 L2=: 3 : '(y^5)%5'
 L3=: 3 : '-(y^7)%7'
 L4=: 3 : ' (y^9)%9'
 Lx0=: L1;L2;L3;L4(L:0)
 Lx=: 3 : ' ;("1),. Lx0 y'
```

3　ベルヌイ · ファミリー

スイスのバーゼルの名門一族。科学分野では Jakob(1654-1705),Johann(1667-1748) の兄弟と Johann の息子 Daniel(1700-1782) など。この Johann はオイラーを育てたことでも知られるが兄とも息子とも仲が悪い。

3.1　調和級数は発散する

調和級数が発散することの証明はニコル · オレーム (1323-1382) が最初である。ノルマンディー生まれでパリで活躍した中世の高名な学僧で哲学、天文学数学, 物理、貨幣論と神学。シャルル 5 世にも仕えた。

調和級数

$$1+\frac{1}{2}+\frac{1}{3}+\frac{1}{4}+\frac{1}{5}+\cdots$$

```
'bar' plot % >:i.20
```



ダンハムはヤコブの証明を載せているが省略

```
  a=. 10;100;1000;10000;100000;1000000
     (;a),. ; ([: +/@:% >:@i.) L:0 a
   10 2.92897
  100 5.18738
 1000 7.48547
10000 9.78761
100000 12.0901
  1e6 14.3927
```

A) 証明にはなっていないがこれで収束することはない

ヤコブの調和級数の分割

$$1+\left(\frac{1}{2}+\frac{1}{3}+\frac{1}{4}\right)+\left(\frac{1}{5}+\frac{1}{6}+\cdots\frac{1}{25}\right)+\left(\frac{1}{26}+\frac{1}{27}+\cdots\frac{1}{676}\right)+\left(\frac{1}{677}+\frac{1}{678}+\cdots\frac{1}{458329}\right)+\cdots$$

括弧内の和は 1 より大きい（ヤコブ）

```
 *: 2 5 26 677  NB.square
4 25 676 458329
```

A) この数を J で探す。(2 個づつ表示）すぐにメモリーオーバーになる。

```
 >(<{@>2 _2) {. L:0  ber_1_sub ''
+-------+-------------+
|1 0    |0 1          |
+-------+-------------+
|2 3    |3 4          |
+-------+-------------+
|5 6    |24 25        |
+-------+-------------+
|26 27  |675 676      |
+-------+-------------+
|677 678|458328 458329|
+-------+-------------+
```

```
   ber_1 ''
1 1.08333 1.73262 3.27819 6.51841
```

■Script .

```
ber_1=: 3 : 0
NB. harmonic series
NB. Usage:  u ''
; +/@:% L:0  ber_1_sub ''
)

ber_1_sub=: 3 : 0
NB. Usage: u y is 4
ANS=. <1
BN0=: 2 NB. Base number
for. i. y do.
BN1=: BN0 , *: BN0
```

```
ORDER=: ({. BN1) + i.  >: -/ |. BN1
ANS=. ANS,<ORDER
BN0=: >:{: ORDER
end.
ANS
)
```

■J Grammar　.

%	*Reciprocal*	単項の % は逆数
[:	*Cappedfork*	左から右へ演算させる
;	*Rase*	ボックスをほぐす
,.	*Stich*	ベクトルを縦に連結
{:	*Last*	最後のアイテムを取り出す
{.	*First*	最初のアイテムを取り出す
<	*Box*	*Box* に入れる
*:	*Square*	2 乗

3.2　ヨハンと X^X

Johann と X^X

$$x^x = 1 + lnx + \frac{x^2(lnx)^2}{2} + \frac{x^3(lnx)^3}{2 \times 3} + \frac{x^4(lnx)^4}{2 \times 3 \times 4} + \cdots \tag{18}$$

$$\int_0^1 x^x dx = 1 - \frac{1}{2^2} + \frac{1}{3^3} - \frac{1}{4^4} + \frac{1}{5^5} - \cdots$$

ヨハン・の積分表
$\int x^k(lnx)^{kn}dx$ を求める
現代風な部分積分を用いた漸化式

$$\int x^m(lnx)^n dx = \frac{1}{m+1}x^{m+1}(lnx)^n - \frac{n}{m+1}\int x^m(lnx)^{n-1}dx \tag{19}$$

$m = \quad 1 \Big| 1 \quad = n$ とする

$$\int xlnxdx = \frac{1}{2}x^2lnx - \frac{1}{2}\int xdx = \frac{1}{2}x^2lnx - \frac{1}{4}x^2$$

$m = \quad 2 \Big| 2 \quad = n$ とする

$$\int x^2(lnx)^2dx = \frac{1}{3}x^3(lnx)^2-\frac{2}{3}\int x^2(lnx)dx = \frac{1}{3}x^3(lnx)^2-\frac{2}{3}\left(\frac{1}{3}x^3lnx-\frac{1}{3}\int x^2dx\right) = \frac{1}{3}x^3(lnx)^2-\frac{2}{9}x^3lnx+\frac{2}{27}x^3$$

級数の積分を各項の積分に置き換える

$$\int_0^1 x^x dx = \int_0^1 \left[1 + xlnx + \frac{x^2(lnx)^2}{2} + \frac{x^3(lnx)^3}{2\times 3} + \frac{x^4(lnx)^4}{2\times 3\times 4}\right] dx$$

$$= \int_0^1 dx + \int_0^1 xlnxdx + \frac{1}{2}\int_0^1 x^2(lnx)^2dx + \frac{1}{2\times 3}\int_0^1 x^3(lnx)^3dx + \frac{1}{2\times 3\times 4}\int_0^1 x^4(lnx)^4dx + \cdots$$

Johann の積分表に置き換える

$$\begin{aligned}\int_0^1 x^x dx = x\,\Big|_0^1 \\ + \frac{1}{2}\left(\frac{1}{2}x^2lnx - \frac{1}{4}x^2\right)\Bigg|_0^1 \\ + \frac{1}{2}\left(\frac{1}{3}x^3(lnx)^2 - \frac{2}{9}x^3lnx + \frac{2}{27}x^3\right)\Bigg|_0^1 \\ + \frac{1}{2\times 3}\left(\frac{1}{4}x^4(lnx)^3 - \frac{3}{16}x^4lnx^2 + \frac{6}{64}x^4lnx - \frac{6}{256}x^4\right)\Bigg|_0^1 \\ + \frac{1}{2\times 3\times 4}\left(\frac{1}{5}x^5(lnx)^4 - \frac{4}{25}x^5lnx^3 + \frac{12}{125}x^5lnx^2 - \frac{24}{625}x^5lnx + \frac{24}{3125}x^5\right)\Bigg|_0^1\end{aligned}$$

lnx の項を全部消す。（最初の階乗の逆数も消える）
[*5]

$$\int_0^1 x^x dx = 1 - \frac{1}{4} + \frac{1}{27} - \frac{1}{256} + \frac{1}{3125} - \cdots$$

$$= 1 - \frac{1}{2^2} + \frac{1}{3^3} - \frac{1}{4^4} + \frac{1}{5^5} - \cdots$$

■J で計算　10 項まで求める

```
 0j20 ":  bern_0 10
0.78343051070873848000
```

6 項目からはほぼ 0 となる

[*5] $lim_{x\to+0}x^m(lnx)^n = 0$　ロピタルの定理

```
2 5 $   bern_sub0 10
          1      _0.25    0.037037 _0.00390625 0.00032
_2.14335e_5 1.21427e_6 _5.96046e_8  2.58117e_9  _1e_10
```



```
bern_0=: 3 : '+/ bern_sub0 y '
NB. x^x by Johann
NB. Usage: bern_0 5
bern_sub0 =: 3 : ' % ((#N)$1 _1)*N^ N=.>:i.y'
```

4　オイラー

4.1　オイラーによる微分

オイラーによる微分

オイラーにとっては dx は 0 に付いた塵であり、無視できるものであるが便利であるから用いる。形あるものをどこまでも小さくするとは考えないので明確である。

ニュートンの sin,cos の級数

$$sinz = z - \frac{z^3}{3!} + \frac{z^5}{5!} - \frac{z^7}{7!} + \cdots$$

$$cos = 1 - \frac{z^2}{2!} + \frac{z^4}{4!} - \frac{z^6}{6!} + \cdots$$

z を dx に置き換える

$$sin(dx) = dx - \frac{(dx)^3}{3!} + \frac{(dx)^5}{5!} - \frac{(dx)^7}{7!} + \cdots$$

$$cos = 1 - \frac{(dx)^2}{2!} + \frac{(dx)^4}{4!} - \frac{(dx)^6}{6!} + \cdots$$

微分の高次のベキ乗は無視できる

$$sin(dx) = dx$$

$$cos(dx) = 1$$

$y = sinx$ において

$$\left(\begin{array}{ll} x \text{ を} & x + dx \\ y \text{ を} & y + dy \end{array} \right. \text{に置き換え}$$

$sin(\alpha + \beta) = sin\alpha \cdot cos\beta + cos\alpha \cdot sin\beta$ を用いて

$$y + dy = sin(x + dx) = sinx \cdot cos(dx) + cosx \cdot sin(dx) = sinx + (cosx)dx$$

両辺から $y = sinx$ を引く

$$dy = sinx + (cosx)dx - y = (cosx)dx$$

微分の比（=導関数）

$$\frac{dy}{dx} = \frac{(cosx)dx}{dx} = cosx$$

4.2 オイラーによる積分

オイラーによる積分
できるだけ無限級数に持ち込み、機能的な形を探してまとめる

Example

$$\int_0^x \frac{sin(lnx)}{lnx}$$

無限級数に持ち込む

$$\frac{sin(lnx)}{lnx} = \frac{lnx - \frac{(lnx)^3}{3!} + \frac{(lnx)^5}{5!} - \frac{(lnx)^7}{7!} + \cdots}{lnx}$$

$$= 1 - \frac{(lnx)^2}{3!} + \frac{(lnx)^4}{5!} - \frac{(lnx)^6}{7!} + \cdots$$

無限級数の和の積分を各項別に積分し和をとる

$$\int_0^x \frac{sin(lnx)}{lnx} dx = \int_0^x dx - \frac{1}{3!}\int_0^1 (lnx)^2 dx + \frac{1}{5!}\int_0^1 (lnx)^4 dx - \frac{1}{7!}\int_0^1 (lnx)^6 dx + \cdots$$

$\int_0 x(lnx)^n dx$ はヨハン・ベルヌイが求めている。オイラーは階乗や π を見抜くのが得意であった

$$\int_0 x(lnx)^2 dx = \left[x(lnx)^2 - 2xlnx + 2x\right]_0^1 = 2 = 2!$$

$$\int_0 x(lnx)^4 dx = \left[x(lnx)^4 - 2x(lnx)^3 + 12x(lnx)^2 - 24xlnx + 24x\right]_0^1 = 24 = 4!$$

$$\int_0 x(lnx)^6 dx = 720 = 6!$$

```
   ^.0 1 2
__ 0 0.693147  NB. ln 1 = 0
```

ライプニッツ級数が現れる

$$\int_0^x \frac{sin(lnx)}{lnx}dx = 1 - \frac{2}{3!} + \frac{24}{5!} - \frac{720}{7!} + \cdots = 1 - \frac{1}{3} + \frac{1}{5} - \frac{1}{7} + \cdots$$

```
   ! 2 4 6
2 24 720
```

```
   (! ^:_1) 2 24 720 NB. !の逆を計算
2 4 6
```

$$\int_0^x \frac{sin(lnx)}{lnx}dx = \frac{\pi}{4}$$

4.3 オイラーの π の公式

オイラーの π の公式

$$\pi = 20arctan\frac{1}{7} + 8arctan\frac{3}{79}$$

これを *arctan* の級数を用いて手計算する

$$\approx 20\left[\frac{1}{7} - \frac{\left(\frac{1}{7}\right)^3}{3} + \frac{\left(\frac{1}{7}\right)^5}{5} - \frac{\left(\frac{1}{7}\right)^7}{7} + \frac{\left(\frac{1}{7}\right)^9}{9} - \frac{\left(\frac{1}{7}\right)^{11}}{11}\right] + 8\left[\frac{3}{79} - \frac{\left(\frac{3}{79}\right)^3}{3} + \frac{\left(\frac{3}{79}\right)^5}{5} - \frac{\left(\frac{3}{79}\right)^7}{7} + \frac{\left(\frac{3}{79}\right)^9}{9} - \frac{\left(\frac{3}{79}\right)^{11}}{11}\right]$$

A) オイラーは 1 時間で手計算できると言っている。16 世紀の数学者には半生を捧げる仕事だった

Z）まあスクリプトを作りましょう。これなら 20 分以下

```
   pi_euler ''
3.1415926535 74190000000000000000
```

Exact 円周率 π=3.1415926535 8979323846

A) 下 14 桁までもとめて 10 桁まで正確。

```
pi_euler=: 3 : 0
NB. find pi
NB. Usage: u ''
PMS=: 6$1 _1 NB. +-
NX0=:1r7, (1r7 ^ NO) * % NO=. 3 5 7 9 11
NX1=:3r79, (3r79 ^ NO) * % NO
0j30 ": +/(20 * PMS * NX0),8*PMS*NX1
)
```

A) 音楽のように古典を弾けというダンハムの趣旨に即し、大数学者の頭の中を少しだけ覗いてみよう

■π の級数展開 .

グレゴリーの *arctan* の級数展開 (1671) を念頭に置いて
[*6]

$$arctan x = x - \frac{x^3}{3} + \frac{x^5}{5} - \frac{x^7}{7} + \frac{x^9}{9} - \cdots$$

π のライプニッツ級数。$x = 1$

収束が遅いので x の値をもっと 0 に近づける

$$\frac{\pi}{4} = arctan(1) = 1 - \frac{1}{3} + \frac{1}{5} - \frac{1}{7} + \frac{1}{9} - \cdots$$

収束の早い関数を探す。*tan* の加法定理から出発する

$$tan(\alpha - \beta) = \frac{tan\alpha - \tan\beta}{1 - tan\alpha \tan\beta}$$

$$tan\alpha = \frac{x}{y}$$
$$tan\beta = \frac{z}{w}$$

$$arctan\frac{x}{y} - arctan\frac{z}{w} = arctan\frac{\frac{x}{y} - \frac{z}{w}}{1 + \frac{x}{y}\frac{z}{w}}$$

整理して

$$arctan\frac{x}{y} = arctan\frac{z}{w} + arctan\frac{xw - yz}{yw + xz}$$

$$\begin{matrix} x = & 1 & 1 & = y \\ w = & 2 & 1 & = z \end{matrix}$$ と置くと

[*6] James Gregory(1638-1675) スコットランド/数学天文学　ライプニッツより数年早かった

$$\frac{\pi}{4} = arctan(1) = arctan\frac{1}{2} + arctan\frac{1}{3}$$

$\begin{array}{rccl} x = & 1 & 2 & = y \\ w = & 7 & 1 & = z \end{array}$ と置くと

$$arctan\frac{1}{2} = arctan\frac{1}{7} + arctan\frac{5}{15}$$

$\begin{array}{rccl} x = & 1 & 3 & = y \\ w = & 7 & 1 & = z \end{array}$ と置くと

$$arctan\frac{1}{3} = arctan\frac{1}{7} + arctan\frac{2}{11}$$

$\begin{array}{rccl} x = & 2 & 11 & = y \\ w = & 7 & 1 & = z \end{array}$ と置くと

$$arctan\frac{2}{11} = arctan\frac{1}{7} + arctan\frac{3}{79}$$

$\frac{\pi}{4} = arctan\frac{1}{2} + arctan\frac{1}{3}$ に代入して整理する

$$\pi = 20arctan\frac{1}{7} + 8arctan\frac{3}{79}$$

Z)18 世紀中頃の出来事。オイラーの鼻歌が聞こえてきそうですが、この頃ロシア音楽なんてあったかな。

A) イングランド時代のヘンデルがよさそう。

4.3.1 江戸初期の円周率の計算

Z）17 世紀後半から 18 世紀初頭の江戸の数学は正に世界のトップクラスであった。漢字の文献でフィードバックできなかったし、ポルトガルやオランダは戦争や金儲けの方が得意だ。

A) 沖方丁「天地明察」に安井算哲=渋川春海が水戸光圀に天文の蘭学書を所望する場面がある

A) 関は現代の Aitken 加速法を用いていたし建部は Richardson 加速を用いている。Aitken と言えばハーバードの計算科学者で K.E.Iverson の先生である。

EXACT　円周率

$$\pi = 3.1415926535\ 8979323846\ 2643383279\ 5028841971\ 6939937510$$

関 (1681) の円周率 。[*7]

$$\pi = 3.14159265359 \text{ 微弱}$$

[*7] 26 桁まで求めたが記述しなかったようである（長田）

建部賢弘 (1722) の円周率

$$pi = 3.1415926535\ 8979323846\ 2643383279\ 5028841971\ 2\ \text{強}$$

関や建部達和算家は定周（零約術を用いて円周率の近似分数を求める際に基準とする円周率）を求めていたので零約術での精度により円周率の必要桁数が決まる。

関の定周

$$\frac{355}{113}$$

建部の定周

$$\frac{5419351}{1725033}$$

```
   0j20 ": _1 x:  ,.355r113 5419351r1725033
3.1415929 2035398250000  NB. Seki
3.141592653589 81530000  NB. Tatebe
```

4.4　ガンマ関数

オイラーの公式

$$[x] = \int_0^1 (-lnt)^x dt$$

n が自然数のとき

$$[x] = \int_0^1 (-lnt)^x dt = n!$$

1729 年のゴールドバッハへの手紙に

$$\frac{1\cdot 2^x}{1+x} \times \frac{2^{1-x}\cdot 3^x}{2+x} \times \frac{3^{1-x}\cdot 4^x}{3+x} \times \frac{4^{1-x}\cdot 5^x}{4+x} \times \cdots$$

整数を入れると階乗が現れる

$$[1] = \frac{1\cdot 2}{2} \times \frac{1\cdot 3}{3} \times \frac{1\cdot 4}{4} \times \frac{1\cdot 5}{5} \times \cdots = 1$$

$$[2] = \frac{1\cdot 2\cdot 2}{3} \times \frac{3\cdot 3}{2\cdot 4} \times \frac{4\cdot 4}{3\cdot 5} \times \frac{5\cdot 5}{4\cdot 6} \times \cdots = 2$$

$$[3] = \frac{1\cdot 2\cdot 2\cdot 2}{4} \times \frac{3\cdot 3\cdot 3}{2\cdot 2\cdot 5} \times \frac{4\cdot 4\cdot 4}{3\cdot 3\cdot 6} \times \frac{5\cdot 5\cdot 5}{4\cdot 4\cdot 7} \times \frac{6\cdot 6\cdot 6}{5\cdot 5\cdot 8} \times \cdots = 6$$

$x = \frac{1}{2}$ とした

$$\left[\frac{1}{2}\right] = \frac{1\cdot\sqrt{2}}{\frac{3}{2}} \times \frac{\sqrt{2}\cdot\sqrt{3}}{\frac{5}{2}} \times \frac{\sqrt{3}\cdot\sqrt{4}}{\frac{7}{2}} \times \frac{\sqrt{4}\cdot\sqrt{5}}{\frac{9}{2}} \times \cdots$$

$$= \sqrt{\frac{2\cdot 4}{3\cdot 3} \times \frac{4\cdot 6}{5\cdot 5} \times \frac{6\cdot 8}{7\cdot 7} \times \frac{8\cdot 10}{9\cdot 9} \times \cdots} \tag{20}$$

ここからオイラーは π を出してくるところが鋭い

$$\left[\frac{1}{2}\right] = \sqrt{\frac{\pi}{4}} = \frac{1}{2}\sqrt{\pi}$$

オイラーの公式

$$[x] = \int_0^1 (-lnt)^x dt$$

対岸までは時間がかかったが辿り着いた

$$\left[-\frac{1}{2}\right] = 2 \times \left[\frac{1}{2}\right] = \sqrt{\pi}$$

■J の階乗とガンマ関数　J の階乗は小数も含まれている。

```
  -:&%: 1p1 NB. Exact 1/2 sqrt(pi)
0.886227
```

Z)(20）をテストしてもなかなか収束しないし簡単に π が出てくる訳でもない。オイラーも苦労したかな

```
   gamma_test 3
+--------+----------------------+
|0.914286|0.888889 0.96 0.979592|
+--------+----------------------+
   gamma_test 6
+--------+-------------------------------------------------+
|0.902181|0.888889 0.96 0.979592 0.987654 0.991736 0.994083|
```

```
+--------+------------------------------------------------+
   gamma_test 10
+---------+
|0.896352|
+---------+
   gamma_test 20
+---------+
|0.891517|
+---------+
   gamma_test 100
+---------+
|0.887324|
+---------+
```

A) ダンハムはオイラーは 1655 年のジョン・ウオリスの深遠な公式を目にしたのではないかと推測している

$$\frac{3\cdot 3\cdot 5\cdot 5\cdot 7\cdot 7\cdot 9\cdot 9\cdots}{2\cdot 4\cdot 4\cdot 6\cdot 6\cdot 8\cdot 8\cdot 10\cdots}=\frac{4}{\pi}$$

■Script J のガンマ関数 ! y は $\Gamma(1+y)$

```
gamma_test=: 3 : 0
NB. euler series for Gamma function
NB. usage:  gamma_test 6
TMPA=:{."1 (;(# tmp),2) $ tmp=:  3<\ 2+i.3 * y
BUNBO=: ; */ (L:0) 2 # L:0  y{. 1{ L:0 TMPA
BUNSI=: y{.0 2 { L:0 TMPA
TMPB=: (;*/ L:0 BUNSI) %  BUNBO
(%: */ TMPB);TMPB
)
```

■ルジャンドルのフォーム 今日のガンマ関数はルジャンドルのフォームが用いられる

$y=-lnt$ と変数変換し、1 ずらす

$$\Gamma(x)=[x-1]=\int_0^{\infty} y^{x-1}e^{-y}dy$$

References

W. ダンダム/一楽 · 実川訳「微積分名作ギャラリー」日本評論社 2009
Brain Bradie [Numerical Analysis] Pearson 2006
Web Page
長田直樹「関孝和の円周率の計算」
平野拓一「数値微分と数値積分」

Miscellances

J602 Download avalable
`http://www.jsoftware.com`
for WIN32/64 Linux32/64 Mac/PPC/Intel PocketPC
Script is in
`http://japla.sakura.ne.jp` J Workshop Oct./2010